Home   Package List   Routine Alphabetical List   Global Alphabetical List   FileMan Files List   FileMan Sub-Files List   Package Component Lists   Package-Namespace Mapping
Routine: DG5300PT

# DG5300PT.m

Go to the documentation of this file.
```DG5300PT ;ALB/SEK NO LONGER APPLICABLE STATUS CLEANUP POST-INS ; 07/31/96
```
``` ;;5.3;Registration;**100**;Aug 13, 1993
```
``` ;
```
``` ;This routine will be run as a post-installation for patch DG*5.3*100.
```
``` ;This routine will change the STATUS field (#.03) to 10 in the
```
``` ;ANNUAL MEANS TEST file (408.13) for any copay test that has a NO
```
``` ;LONGER REQUIRED DATE (#.17) and the STATUS field of 7.  This
```
``` ;problem only occurred when the patient went from non-exempt SC less
```
``` ;than 50% to SC 50% to 100%.
```
``` ;
```
```POST ;entry point for post-install, setting up checkpoints
```
``` N %
```
``` S %=\$\$NEWCP^XPDUTL("DGTTDT","EN^DG5300PT",0)
```
``` Q
```
``` ;
```
```EN ; begin processing
```
``` ;
```
``` ;go through ANNUAL MEANS TEST file changing STATUS to 10 for copay
```
``` ;tests that have a NO LONGER REQUIRED DATE and STATUS of 7.
```
``` N DGTTDT
```
``` ;
```
``` D BMES^XPDUTL("  >> Copay no longer applicable status Clean-up")
```
``` ;
```
``` ;get value from checkpoints, previous run
```
``` S DGTTDT=+\$\$PARCP^XPDUTL("DGTTDT")
```
``` ;
```
```LOOP ;
```
``` N DFN,DGFL,DGFLD,DGIEN,DGMTA,DGVAL,%
```
``` F  S DGTTDT=\$O(^DGMT(408.31,"AS",2,7,DGTTDT)) Q:'DGTTDT  D
```
``` .S DFN=0 F  S DFN=\$O(^DGMT(408.31,"AS",2,7,DGTTDT,DFN)) Q:'DFN  D
```
``` ..S DGIEN=0 F  S DGIEN=\$O(^DGMT(408.31,"AS",2,7,DGTTDT,DFN,DGIEN)) Q:'DGIEN  D
```
``` ...S DGMTA=\$G(^DGMT(408.31,DGIEN,0)) Q:'DGMTA
```
``` ...Q:\$P(DGMTA,"^",17)=""
```
``` ...S DGFL=408.31,DGFLD=.03,DGVAL=7 D KILL^DGMTR
```
``` ...S DGVAL=10,\$P(^DGMT(408.31,DGIEN,0),"^",3)=DGVAL D SET^DGMTR
```
``` .;update checkpoint
```
``` .S %=\$\$UPCP^XPDUTL("DGTTDT",DGTTDT)
```
``` Q
```
{"ft_lang_label":"__label__en","ft_lang_prob":0.60812306,"math_prob":0.9021593,"size":3331,"snap":"2021-31-2021-39","text_gpt3_token_len":1343,"char_repetition_ratio":0.11571987,"word_repetition_ratio":0.32397407,"special_character_ratio":0.3725608,"punctuation_ratio":0.23300971,"nsfw_num_words":0,"has_unicode_error":false,"math_prob_llama3":0.96872985,"pos_list":[0],"im_url_duplicate_count":[null],"WARC_HEADER":"{\"WARC-Type\":\"response\",\"WARC-Date\":\"2021-07-29T11:20:52Z\",\"WARC-Record-ID\":\"<urn:uuid:3e83f353-c0fa-4f6a-b01b-d69500020dca>\",\"Content-Length\":\"20883\",\"Content-Type\":\"application/http; msgtype=response\",\"WARC-Warcinfo-ID\":\"<urn:uuid:b452db68-d6cf-4ce1-98cc-bcbb7071cbca>\",\"WARC-Concurrent-To\":\"<urn:uuid:4f48f79b-1537-409e-a2fa-573e49515965>\",\"WARC-IP-Address\":\"156.96.113.49\",\"WARC-Target-URI\":\"https://vivian.worldvista.org/dox/Routine_DG5300PT_source.html\",\"WARC-Payload-Digest\":\"sha1:PZU24O6VZO2YA4VXW6TGC37ZBTOQKMDL\",\"WARC-Block-Digest\":\"sha1:K2TMH43ZV3FAGAAOJWOWIGEYPS5VUQ4B\",\"WARC-Identified-Payload-Type\":\"text/html\",\"warc_filename\":\"/cc_download/warc_2021/CC-MAIN-2021-31/CC-MAIN-2021-31_segments_1627046153857.70_warc_CC-MAIN-20210729105515-20210729135515-00708.warc.gz\"}"}